UNIVERSITAT POLITÈCNICA DE CATALUNYA
BARCELONATECH
Departament d'Estadística i Investigació Operativa

Programació Lineal i Entera, curs 2016-17
2on curs Grau en Estadística UB-UPC
Examen final de laboratori, 12/06/2017

(Part 1). Una empresa constructora està planificant els moviments de maquinària pesant des d'uns punts (A, B i C) als propers llocs on s'iniciaran obres (X, Y i Z). Es precisen 5 màquines al lloc X, 4 a Y i 3 a Z, i actualment hi ha 8 al punt A, 5 a B i 3 a C. El cost de traslladar una màquina entre els emplaçaments es mostra a la taula següent:

(unitats monetàries)	X	Y	Z
A	50	60	30
B	60	40	20
C	40	70	30

a) **(4 pts.) Formuleu matemàticament** el problema de programació completament parametritzat que permet trobar els moviments a realitzar a cost mínim: indiqueu quins son el paràmetres del problema i definiu <u>formalment</u> les variables de decisió, les constriccions i la funció objectiu.

O={A, B, C}: Conjunt d'índexs de orígens
D={X, Y, Z}: Conjunt d'índexs de destinacions
a=[8, 5, 3]: vector de capacitats dels orígens
b=[5, 4, 3]: vector de demandes de les destinacions

Variables de decisió:
x_{ij}: nombre de maquines transportades des de l'origen $i \in O$ a la destinació $j \in D$. $\forall i \in O$, $\forall j \in D$

Constriccions:

(1) $\sum_{j \in D} x_{ij} \leq a_i \qquad \forall i \in O$ Des de cada origen no surten més màquines que les disponibles.

(2) $\sum_{i \in O} x_{ij} \geq b_j \qquad \forall j \in D$ A cada destinació arriben les màquines necessàries.

Funció Objectiu: $\sum_{i \in O} \sum_{j \in D} c_{ij} x_{ij}$ Minimitzar els costos totals de transport.

b) **(2 pt.) Resoleu** amb OPTMODEL el problema anterior i presenteu la solució final amb el cost que suposa.

```
proc optmodel presolver = 0;

/* Paràmetres */
set<string> PTS_ACTUAL = {'A', 'B', 'C'};
set<string> PTS_FUTUR = {'X', 'Y', 'Z'};

number demanda {PTS_FUTUR} = [5 4 3];
number oferta {PTS_ACTUAL} = [8 5 3];
number cost {PTS_ACTUAL, PTS_FUTUR} =
         [ 5060     30
           6040     20
           4070     30 ];

/* Variables */
var mov {PTS_ACTUAL, PTS_FUTUR} >=0 integer;

min Total_Cost = sum {i in PTS_ACTUAL, j in PTS_FUTUR} cost[i, j]*mov[i, j];

con Dem {j in PTS_FUTUR} :
      sum {i in PTS_ACTUAL} mov[i,j] >= demanda[j];

con Of {i in PTS_ACTUAL} :
      sum {j in PTS_FUTUR} mov[i,j] <= oferta[i];
```

```
solve with MILP;

print mov;
```

Cost al òptim: 460 u.m.

mov			
	X	**Y**	**Z**
A	2	0	2
B	0	4	1
C	3	0	0

(Part 2). Considerem una variant del problema anterior. Ara, la demanda de maquinària als llocs X, Y i Z ha augmentat a 10, 7 i 8 respectivament. Per enfrontar-se a l'increment de demanda, la empresa ha decidit recórrer a un proveïdor extern, que col·laborarà conjuntament amb A, B i C en els moviments de maquinària pesant fins a X, Y, Z. En concret, la empresa ha decidit contractar un únic proveïdor extern, d'entre dos possibles candidats (P i Q). Qualsevol d'aquest possibles proveïdors tindria capacitat suficient per a subministrar les màquines necessàries. El contracte amb el proveïdor P tindria un cost fixe de 300 u.m., i el del proveïdor Q un cost fixe de 250 u.m.

Els costos de trasllat de maquinària associats a aquests proveïdors són:

(unitats monetàries)	X	Y	Z
P	55	60	75
Q	80	65	45

c) **(2.5 pts.) Formuleu matemàticament** el problema de programació completament parametritzat que afegeix les condicions anteriors al problema de la primera part, destacant formalment els canvis: indiqueu, si n'hi han, les noves variables de decisió i les noves constriccions, i/o les modificacions al primer model.

Dades addicionals:
Conjunt d'índexs de possibles proveïdors NO={P, Q}
Cost fixe de contracte amb proveïdor $i \in NO$: f_i
Com que cada possible proveïdor pot subministrar tantes màquines com siguin necessàries, definim la capacitat de cadascú d'ells com la suma de totes les demandes. Es a dir, $\tilde{a}_i = \sum_{j \in D} b_j$, for all $i \in NO$.

Formulació:

Noves variables de decisió:

x_{ij}: nombre de maquines transportades des de el proveïdor $i \in NO$ a la destinació $j \in D$. $\forall i \in NO$, $\forall j \in D$
$y_i \in \{0,1\}$, $\forall\ i \in NO$. $y_i = 1 \Leftrightarrow$ es contracta el proveïdor $i \in NO$.

Les constriccions (1) no canvien.
Les constriccions (2) es transformen en:

(2') $\sum_{i \in O \cup NO} x_{ij} \geq b_j \qquad \forall j \in D \qquad$ A cada destinació arriben les màquines necessàries.

Apareixen els següents conjunts de constriccions associades possibles proveïdors:

$\sum_{i \in NO} y_i = 1 \qquad$ Es selecciona exactament un proveïdor.

$\sum_{j \in D} x_{ij} \leq \tilde{a}_i y_i \qquad \forall i \in NO \qquad$ Relació x$\leftrightarrow$y

(si no es contracta el proveïdor i, no pot repartir màquines.)

La nova Funció Objectiu:

$$\sum_{i \in O \cup NO} \sum_{j \in D} c_{ij} x_{ij} + \sum_{i \in NO} f_i y_i$$

Minimitzar la suma dels costos totals de transport més el cost del contracte del proveïdor extern.

d) **(1.5 pts.) Resoleu** amb OPTMODEL el problema anterior i presenteu la solució final amb el cost que suposa.

```
proc optmodel presolver = 0;

/* Paràmetres */
set<string> PTS_ACTUAL = {'A', 'B', 'C'};
set<string> PTS_FUTUR = {'X', 'Y', 'Z'};
set<string> PROVS = {'P', 'Q'};

number demanda {PTS_FUTUR} = [10 7 8];
number oferta {PTS_ACTUAL} = [8 5 3];
number cost {PTS_ACTUAL, PTS_FUTUR} =
         [ 50 60    30
           60 40    20
           40 70    30 ];
number cost_prov {PROVS, PTS_FUTUR} =
         [ 55 60    75
           80 65    45];
number fix_cost {PROVS} = [300 250];

/* Variables */
var mov {PTS_ACTUAL, PTS_FUTUR} >=0 integer;
var mov_pr {PROVS, PTS_FUTUR} >=0 integer;
var cont {PROVS} binary;

min Total_Cost = sum {i in PTS_ACTUAL, j in PTS_FUTUR} cost[i, j]*mov[i, j]
               + sum {k in PROVS, j in PTS_FUTUR} cost_prov[k, j]*mov_pr[k, j]
               + sum {k in PROVS} fix_cost[k]*cont[k];

con Dem {j in PTS_FUTUR} :
      sum {i in PTS_ACTUAL} mov[i,j] + sum {k in PROVS} mov_pr[k, j] >= demanda[j];

con Of {i in PTS_ACTUAL} :
      sum {j in PTS_FUTUR} mov[i,j] <= oferta[i];

con coupling {k in PROVS} :
      sum {j in PTS_FUTUR} mov_pr[k,j] <= cont[k] * sum {j in PTS_FUTUR} demanda[j];

con choice : sum {k in PROVS} cont[k] = 1;

solve with MILP;

print mov;
print mov_pr;
```

Cost al òptim: 1365 u.m. El proveïdor escollit és el P:

mov	**X**	**Y**	**Z**
A	0	0	8
B	0	5	0
C	3	0	0

mov_pr	**X**	**Y**	**Z**
P	7	2	0
Q	0	0	0